Z-100 LifeLine
Since 1989

March 2022

#WEB

This article was first published in issue #69, June 2000

ZVIDEO.COM Documentation

by Steven Vagts
Editor, "Z-100 LifeLine"

ZVIDEO.COM

New for MS-DOS v4.01, **ZVIDEO** allows us to save and restore our favorite Z-100 video configuration. This becomes very useful when an application program leaves us with the wrong screen color or incorrect number of lines displayed on the screen.

This became particularly apparent with the HELP system included with MS-DOS 4. The standard Z-100 display has 24 lines per screen, but for HELP to work properly, the display needs 25 lines.

John Beyers wrote the following routine to solve this inconsistency. This routine can then be placed in a batch file to save the video configuration before running the application, then run again after the application is terminated to restore the configuration.

There are many ways to save and restore the Z-100 video configuration. But the first task is to find some memory location to save the current video characteristics.

Where do we have some free memory? Nowhere that is absolutely safe, unless we tell DOS about it by creating a TSR (Terminate but Stay Resident) program. Since it would be nice to NOT have to do that, what is the next best choice?

Well, unused interrupts have 4 bytes (32 bits) that could be used as temporary storage, as long as we can be reasonably sure that an interrupt will not be called.

Because Int 5Fh is reserved in the "PC" world and is not being used by any Z-100 BIOS routines, it was chosen. However, if a problem is found later, this interrupt can be changed at compile time by simply setting the equate SVintNUM to a different Hex value.

You also need to realize that the original value of the Int # chosen will be permanently altered (destroyed) until a re-boot is performed. Again, this should not be a problem unless some other program wants to also steal that memory for some unknown function.

The next step is to decide what video characteristics to save/restore and how many bits each will take. We can add more - up to 32 bits total. Here is the current list and the # of bits each takes:

Reverse Video Condition	1
Foreground Color (0-7)	3
Background Color (0-7)	3
Graphics Mode Condition	1
Cursor Shape (Blink/Block/Underline)	8
Displayed Lines Per Screen (Max 50)	6
Status Line Condition	1
Line Wrap Condition	1
Key Click Condition	1
For a Total of	25 bits

Last, we must decide how to interface to the user. This program will require a command line switch, either /s for save or /r for restore, or it will prompt the user with a help screen.

TITLE **ZVIDEO** - Save or Restore Z100 Video Characteristics
; by John Beyers, 5/2000

```
INCLUDE PARMS.ASM
INCLUDE VER.ASM
INCLUDE DEFBIOS.ASM
INCLUDE DEFMTR.ASM
INCLUDE DEFCONFG.ASM
INCLUDE DEFMS.ASM
INCLUDE DEFDOSI.ASM
INCLUDE DEFASCII.ASM
```

```
INCLUDE DEFIPAGE.ASM
INCLUDE DEFPSP.ASM

SVintNUM = 5Fh

DUMMY   SEGMENT  STACK
  ;Prevent Link Error Message
DUMMY   ENDS

CODE   SEGMENT
       ASSUME        CS:CODE,DS:CODE,ES:CODE,SS:CODE

       ORG    100h
START:        JMP    BEGIN  ;Skip over data area
HELP   DB     CC_CR,CC_LF
DB 'This program can be used to save and then
   restore some of the Z100 Video',CC_CR,CC_LF
DB 'Characteristics that sometimes get inadver-
   tantly changed by programs.',CC_CR,CC_LF
DB 'Valid command line switches are:
   ',CC_CR,CC_LF
DB '   /S     - Save the current settings.',
   CC_CR,CC_LF
DB '   /R     - Restore the last saved settings.',
   CC_CR,CC_LF
DB 'Anything else on the command line will generate
   this screen. Please see the',CC_CR,CC_LF
DB 'source code for additional information.',
   CC_CR,CC_LF,CC_LF
DB '<CTRL-C> to Exit, (S)ave or (R)estore ? '
HELPLEN = $-HELP

RESETmtr      DB CC_ESC,'m'
FOREcolor     DB '0'
BACKcolor     DB '0',CC_ESC
STATUSline    DB 'y1',CC_ESC
CLICK         DB 'x2',CC_ESC,'y5'  ;Also Enable Cursor
RESETmtrLEN = $-RESETmtr

CONgioctl LABEL NEAR
       DB 0   ;Level
       DB 0   ;Reserved
       DW 14  ;Length of following data
       DW 0   ;Control flags(bit 0 - Intensity/Blink)
       DB 1   ;Mode type 1-text, 2=graphics
       DB 0   ;Reserved
       DW 3   ;Color/Mono - Bits per pixel
       DW 8   ;Pixel Columns
       DW 9   ;Pixel Rows
       DW 80  ;Character Columns
LPS    DB 24,0       ;Character Rows

BEGIN:
       MOV    SI,pspCommandTail
       CALL   SOB0
CHKINP:
       CALL   MCU
       CMP    AL,'R'               ;Restore?
       JZ     RESTORE
       CMP    AL,'S'               ;Save?
       JZ     SAVE
       MOV    BX,STDERR     ;Standard Error Device
       MOV    CX,HELPLEN
       MOV    DX,offset HELP   ;Ask what to do
       SCALL  WRITEH
       MOV    AL,DOSF_CONIN        ;Console input
       SCALL  CONINF
                     ;Flush keybrd buffer & input
       JMP    CHKINP
SAVE:
       CALL   GETMTRDS
ASSUME ES:MTR_D_SEG
       MOV    DL,ES:MTR_lns_per_scrn
       MOV    DH,ES:MTR_msk
                        ;Fore&Background Colors
       MOV    AH,ES:MTR_cursor_value
                        ;Blink/Block/Underline
       MOV    AL,ES:MTR_key_click_flg
       ROR    ES:MTR_line_25_flg,1
       RCL    AL,1
       ROR    ES:MTR_grphc_flg,1
       RCL    AL,1
       ROR    ES:MTR_rvrs_video_flg,1
       RCL    AL,1
       ROR    ES:MTR_wrp_ln_flg,1
       RCL    AL,1
       MOV    DS,AX
       MOV    AL,SVintNUM
       SCALL  SIVEC    ;Set Int Vec to DS:DX
       JMP short QUIT
RESTORE:
       MOV    AL,SVintNUM
       SCALL  GIVEC    ;Get Int Vec in ES:BX
       MOV    DX,ES    ;Saved values in DX and BX
       CALL   GETMTRDS
ASSUME ES:MTR_D_SEG
       MOV    ES:MTR_cursor_value,DH
                        ;Blink/Block/Underline
       SHR    DL,1
       SBB    AL,AL
       MOV    ES:MTR_wrp_ln_flg,AL
       SHR    DL,1
       SBB    AX,AX
       MOV    ES:MTR_rvrs_video_flg,AX
       SHR    DL,1
       SBB    AL,AL
       MOV    ES:MTR_grphc_flg,AL
       SHR    DL,1
       SBB    AL,AL
       ADD    STATUSline,AL
       SHR    DL,1
       SBB    AL,AL
       SUB    CLICK,AL
       MOV    AL,BH
       AND    AL,111b
       ADD    FOREcolor,AL
       MOV    AL,BH
       MOV    CL,3
       SHR    AL,CL
       AND    AL,111b
       ADD    BACKcolor,AL
       PUSH   BX
       MOV    BX,STDERR     ;Standard Error Device
       MOV    CX,RESETmtrLEN
       MOV    DX,Offset RESETmtr
       SCALL  WRITEH
       CALL   GET
       POP    AX
       DEC    AX
       MOV    LPS,AL
       CALL   SET

QUIT:  SCALL  EXIT

;*     MCU - Map character to upper

MCU:   CMP    AL,'a'
       JC     MCU1
       CMP    AL,'z'+1      ;In range?
       JNC    MCU1
       SUB    AL,'a'-'A'    ;Yes, map it
MCU1:  RET

;*     SOB - Skip Over Blanks

SOB:   CMP    AL,'/'        ;Switch Character?
       JZ     SOB0
       CMP    AL,CC_HT      ;Tab?
       JZ     SOB0
       CMP    AL,CC_SP      ;Space?
       JNZ    SOB1
SOB0:  LODSB
       JMP    SOB
SOB1:  RET

GETMTRDS:
       XOR    AX,AX
       MOV    ES,AX
ASSUME  ES:IPAGE_SEG
       MOV    AX,ES:MTR_DS
       MOV    ES,AX    ;Set ES to MTR ROM Data Segmnt
       RET
```

```
GET:   MOV    CL,7FH
SETCONT:
       MOV    AX,440CH
       MOV    BX,STDOUT
       MOV    CH,3
       MOV    DX,offset CONgioctl
       INT    21H
       RET
SET:   MOV    CL,5FH
       JMP    SETCONT

CODE   ENDS
       END    START
```

This program and other updated Z-100 MS-DOS v4 software are available from "*Z-100 LifeLine*".

The ZVIDEO program can be used with any programs that require a certain environment to operate. It simply resets the Z-100 environment upon completion. For example, before running the ZDOS v4 HELP system, which requires 25 lines per screen, rather than the standard 24, run the command:

ZVIDEO /s

to save the present Z-100 environment. Set the needed 25 lines per screen with the command:

SETLPS 25

Then when you are done with HELP, run the command:

ZVIDEO /r

to reset the original Z-100 environment.

But we can also run all this from a batch file. For example:

The batch file **HELPZ.BAT** can be used to invoke HELP without running ZPC. The batch file consists of:

```
@ECHO OFF
ZVIDEO /S
REM ZVIDEO /S saves the video parameters.
SETLPS 25
REM Sets the Z-100 video screen for 25
    lines/screen.
CLS
ECHO  * * Z-100 MS-DOS HELP PROGRAM * *
ECHO.
ECHO Note: CTRL is the {F0} key and the
     keypad cannot be shifted in
ECHO advance, so the key assignments are:
ECHO   PgUp is {DEL LINE}  or
               {SHIFT}{keypad 9}.
ECHO   PgDn is {INS LINE}  or
               {SHIFT}{keypad 3}.
ECHO   CTRL-END is {F0}-{SHIFT}{HOME} or
ECHO           {F0}-{SHIFT}{keypad 1} or
ECHO           press {ESC} to exit.
ECHO.
ECHO Note: To shift/unshift the keypad,
     press {LINE FEED}.
ECHO       To go to 1st page of a multi-
           page document, press {HOME}.
ECHO       To go to the last page,
           press {SHIFT}{HOME}.
ECHO.
PAUSE
HELP %1 %2
ZVIDEO /R
REM ZVIDEO /R resets video parameters.
CLS
```

And to remedy the lines per screen problem under ZPC, I've written another batch file, **HELPPC.BAT**, to invoke HELP:

```
@ECHO OFF
ZVIDEO /S
REM ZVIDEO /S saves the video parameters.
SETLPS 25
REM Sets the Z-100 video screen for 25
    lines/screen.
CLS
ECHO  * * Z-100 PC-DOS HELP PROGRAM * *
ECHO.
ECHO Note: The keypad is already shifted,
     so the key assignments are:
ECHO   PgUp is {DEL LINE}  or
               {keypad 9}.
ECHO   PgDn is {INS LINE}  or
               {keypad 3}.
ECHO   CTRL-END is {CTRL}-{SHFT}{HOME} or
ECHO              {CTRL}-{keypad 1} or
ECHO              press {ESC} to exit.

ECHO.
ECHO Note: To shift/unshift the keypad,
     press {HELP}.
ECHO       To go to first page of a
           multipage document, press
           {HOME}.
ECHO       To go to the last page,
           press {SHIFT}{HOME}.
ECHO.
PAUSE
HELP %1 %2
ZVIDEO /R
CLS
```

Note: ESCape sequences used in ZVIDEO.COM are not recognized while running HELP under ZPC, so only the lines per screen are reset upon exiting.

I hope you find this article helpful.
If you have any questions or comments, please email me at:
z100lifeline@swvagts.com

Cheers,

Steven W. Vagts